```
In[44]:= ClearAll["Global`*"]

(*Probabilities of selection for the case sR is played and for the case sL*)

pR[sR_] = 1 / 2 + sR;
pL[sL_] = 1 / 2 + sL;

(*We compute the continuation values for strategies (sL, sR, xL, xR)*)

(* We use the following notation for the policies: we consider x_i,
the "concession" of an agent i,
hence L's policy y_L=  0 + x_L and R's policy y_R= 1 - x_R*)

U = Solve[
   wLl == (1 - pL[sL]) * (b - xL + beta * wLl ) + pL[sL] * (- (1 - xR) + beta * wLr) &&
    wLr == (1 - pR[sR]) * (b - xL + beta * wLl ) + pR[sR] * (- (1 - xR) + beta * wLr) &&
    wRl == (1 - pL[sL]) * (- (1 - xL) + beta * wRl ) + pL[sL] * (b - xR + beta * wRr) &&
    wRr == (1 - pR[sR]) * (- (1 - xL) + beta * wRl ) + pR[sR] * (b - xR + beta * wRr) &&
    wPl ==
     (1 - pL[sL]) * (- (theta - xL) + beta * wPl ) + pL[sL] * (- (1 - xR - theta) + beta * wPr) &&
    wPr == (1 - pR[sR]) * (- (theta - xL) + beta * wPl ) +
      pR[sR] * (- (1 - xR - theta) + beta * wPr), {wLl, wLr, wRl, wRr, wPl, wPr}];

(*R's continuation value when R was selected last*)
wRr[beta_, b_, sL_, sR_, xL_, xR_] = FullSimplify[ wRr /. U[[1]]];
(*L's continuation value when L was selected last*)
wLl[beta_, b_, sL_, sR_, xL_, xR_] = FullSimplify[ wLl /. U[[1]]];
(*Principal's continuation value when L was selected last*)
wPl[beta_, b_, theta_, sL_, sR_, xL_, xR_] = FullSimplify[ wPl /. U[[1]]];
(*Principal's continuation value when R was selected last*)
wPr[beta_, b_, theta_, sL_, sR_, xL_, xR_] = FullSimplify[ wPr /. U[[1]]];

(**Principal's and R's punishment**)

alphaR[m_, beta_, b_] = FullSimplify[ With[{xL = 0}, With[{xR = 0}, (*Polarization*)
    With[{sL = -m}, With[{sR = -m}, (*P's response to polarization*)
      wRr[beta, b, sL, sR, xL, xR] (*R's payoff*)
     ]]]]]
alphaP[m_, beta_, b_, theta_] =
 FullSimplify[ With[{xL = 0}, With[{xR = 0}, (*Polarization*)
    With[{sL = -m}, With[{sR = -m}, (*P's response to polarization*)
      wPl[beta, b, theta, sL, sR, xL, xR] (*P's payoff*)
     ]]]]]
Out[52]=
```

$$\frac{1 - b + 2\,(1 + b)\,m}{2\,(-1 + beta)}$$

Out[53]=

$$\frac{1+m\,(-2+4\,\text{theta})}{2\,(-1+\text{beta})}$$

In[54]:=
```
(**Agent's enforcement constraints**)

(*Making use of the fact that the punishment is a
 continuation contract **pretending** that R had been selected last*)
alphaL[m_, beta_, b_, theta_, sL_, sR_, xL_, xR_] = FullSimplify[wLr /. U⟦1⟧]

enforcementLmax[m_, beta_, b_, sL_, sR_, xL_, xR_] =
  FullSimplify[ xL + beta * (alphaL[m, beta, b, theta, sL, sR, xL, xR] - wLl) /. U⟦1⟧];
(*alphaL still to be computed*)
enforcementRmax[m_, beta_, b_, sL_, sR_, xL_, xR_] =
  FullSimplify[ xR + beta * (alphaR[m, beta, b] - wRr) /. U⟦1⟧];
```

Out[54]=

$$\frac{1+b\,(-1+2\,\text{sR})+\text{xL}-2\,\text{beta}\,(\text{sL}-\text{sR})\,(-1+\text{xR})-\text{xR}-2\,\text{sR}\,(-1+\text{xL}+\text{xR})}{2\,(-1+\text{beta})\,(1+\text{beta}\,(\text{sL}-\text{sR}))}$$

In[57]:=
```
(**R' Actions in the Optimal contract**)
(*Optimal contract -- Solving R's enforcement constraints*)

(*Substituting that we know that L's lead is followed by endorsement of L*)
enforcementRmaxsubstituted[m_, beta_, b_, xL_, xR_, sR_] = FullSimplify[With[{sL = -m},
   enforcementRmax[m, beta, b, sL, sR, xL, xR]]]
solvexR = FullSimplify[Solve[enforcementRmaxsubstituted[m, beta, b, xL, xR, sR] == 0, xR]]
(*find the largest concession by R*)
xRmax[m_, beta_, b_, xL_, sR_] = FullSimplify[xR /. solvexR⟦1⟧];
(*store that concession*)
```

Out[57]=

$$\frac{(1+b)\,\text{beta}\,(-2+\text{beta}+2\,\text{beta}\,m)\,(m+\text{sR})+\text{beta}\,(-1+2\,\text{sR})\,\text{xL}-(-2+\text{beta}+2\,\text{beta}\,m)\,\text{xR}}{2\,(-1+\text{beta})\,(-1+\text{beta}\,(m+\text{sR}))}$$

Out[58]=

$$\left\{\left\{\text{xR}\to(1+b)\,\text{beta}\,(m+\text{sR})+\frac{\text{beta}\,(-1+2\,\text{sR})\,\text{xL}}{-2+\text{beta}+2\,\text{beta}\,m}\right\}\right\}$$

In[60]:=
```
(*Finding L's concessions in the Optimal contract, xL*)

enforcementLmaxsubstituted[m_, beta_, b_, xL_, sR_] =
 FullSimplify[With[{xR = xRmax[m, beta, b, xL, sR]},
   With[{sL = -m}, enforcementLmax[m, beta, b, sL, sR, xL, xR]]]]

xLmaxsolve = FullSimplify[Solve[enforcementLmaxsubstituted[m, beta, b, xL, sR] == 0, xL]]
xLmax[m_, beta_, b_, sR_] = FullSimplify[xL /. xLmaxsolve[[1]]];
```

Out[60]=

$$-((1+b)\,\text{beta}\,(m+sR))+\frac{(-2+\text{beta}-2\,\text{beta}\,sR)\,xL}{-2+\text{beta}+2\,\text{beta}\,m}$$

Out[61]=

$$\left\{\left\{xL\to-\frac{(1+b)\,\text{beta}\,(-2+\text{beta}+2\,\text{beta}\,m)\,(m+sR)}{2+\text{beta}\,(-1+2\,sR)}\right\}\right\}$$

In[63]:=
```
(**Obtaining L's Punishment and the Optimal Contract -- we substitute xL**)

wPrOptimalContractprevious[m_, beta_, b_, theta_, sR_] =
 FullSimplify[With[{sL = -m}, With[{xL = xLmax[m, beta, b, sR]},
    With[{xR = xRmax[m, beta, b, xL, sR]}, wPr[beta, b, theta, sL, sR, xL, xR]]]]]
```

Out[63]=

$$\Big(-2+sR\,(-4+8\,\text{theta})+\text{beta}^2\,(m+sR)\,(-3-6\,m+2\,sR+4\,m\,sR+b\,(-1-6\,m-2\,sR+4\,m\,sR)+2\,\text{theta}-4\,sR\,\text{theta})+\text{beta}\,(1+4\,m\,(2+b-\text{theta})+4\,sR\,(2+b-sR+2\,(-1+sR)\,\text{theta}))\Big)\Big/\big(2\,(-1+\text{beta})\,(-1+\text{beta}\,(m+sR))\,(2+\text{beta}\,(-1+2\,sR))\big)$$

In[64]:=
```
DwPrOptimalContractprevious[m_, beta_, sR_] =
 FullSimplify[D[wPrOptimalContractprevious[m, beta, b, theta, sR], b]]
```

Out[64]=

$$\frac{\text{beta}\,(m+sR)\,(4+\text{beta}\,(-1-2\,sR+m\,(-6+4\,sR)))}{2\,(-1+\text{beta})\,(-1+\text{beta}\,(m+sR))\,(2+\text{beta}\,(-1+2\,sR))}$$

In[65]:=
```
wPlOptimalContractprevious[m_, beta_, b_, theta_, sR_] =
 FullSimplify[With[{xL = xLmax[m, beta, b, sR]}, With[{sL = -m},
    With[{xR = xRmax[m, beta, b, xL, sR]}, wPl[beta, b, theta, sL, sR, xL, xR]]]]]
```

Out[65]=

$$\Big(-2+2\,(1+b)\,\text{beta}^3\,(1+2\,m)\,(m+sR)^2+m\,(4-8\,\text{theta})-\text{beta}^2\,(m+sR)\,((1+b)\,(1+4\,m\,(2+m)+4\,sR)+(2-4\,sR)\,\text{theta})+\text{beta}\,(1+2\,sR+4\,sR\,(b+\text{theta})+m\,(2+4\,b+4\,sR+8\,\text{theta}-8\,sR\,\text{theta}))\Big)\Big/\big(2\,(-1+\text{beta})\,(-1+\text{beta}\,(m+sR))\,(2+\text{beta}\,(-1+2\,sR))\big)$$

In[66]:=
```
DwPlOptimalContractprevious[m_, beta_, sR_] =
 FullSimplify[D[wPlOptimalContractprevious[m, beta, b, theta, sR], b]]
```

Out[66]=

$$\frac{\text{beta}\ (m+sR)\ \left(4+2\ \text{beta}^2\ (1+2\ m)\ (m+sR)-\text{beta}\ (1+4\ m\ (2+m)+4\ sR)\right)}{2\ (-1+\text{beta})\ (-1+\text{beta}\ (m+sR))\ (2+\text{beta}\ (-1+2\ sR))}$$

In[67]:=
```
(*Showing that the principal's payoff regardless of who leads
 STRICTLY decreases in b for any interior sR meaning that if some sR is
 feasible for some b it provides higher expected payoffs for larger b's*)
```

In[68]:=
```
(*To Do this we need to condition on the fact that \overline{\theta}<1/2,
that b<bhat && theta>\overlineline theta*)
overlineTheta[m_, beta_, b_] = 2 beta / (1 - beta) (1 - beta (1 / 2 + m)) (b + 1) m
(*from the paper*)
bHat[m_, beta_, theta_] =
 (4 + beta^2 (2 + 8 m (1 + m (-1 + theta)) - 2 theta) + 8 beta (-1 + theta) - 8 theta) /
  (beta (4 + beta (-1 + 4 (-2 + m) m)))
 (*from the other file*)
```

Out[68]=

$$\frac{2\ (1+b)\ \text{beta}\ m\ \left(1-\text{beta}\ \left(\frac{1}{2}+m\right)\right)}{1-\text{beta}}$$

Out[69]=

$$\frac{4+\text{beta}^2\ (2+8\ m\ (1+m\ (-1+\text{theta}))-2\ \text{theta})+8\ \text{beta}\ (-1+\text{theta})-8\ \text{theta}}{\text{beta}\ (4+\text{beta}\ (-1+4\ (-2+m)\ m))}$$

In[70]:=
```
Reduce[DwPrOptimalContractprevious[m, beta, sR] ≤ 0 && 0 < beta < 1 && 0 < m < 1 / 2 &&
  (1 - beta) ≥ 2 m beta && m > sR > -m && overlineTheta[m, beta, b] < 1 / 2 &&
  0 < b < bHat[m, beta, theta] && theta > overlineTheta[m, beta, b]]
```

Out[70]=
```
False
```

In[71]:=
```
Reduce[DwPlOptimalContractprevious[m, beta, sR] ≤ 0 && 0 < beta < 1 && 0 < m < 1 / 2 &&
  (1 - beta) ≥ 2 m beta && m > sR > -m && overlineTheta[m, beta, b] < 1 / 2 &&
  0 <  b < bHat[m, beta, theta] && theta > overlineTheta[m, beta, b]]
```

Out[71]=
```
False
```

In[72]:=
```
(*Show that the distance to the punishment strictly decreases for every sR*)
```

In[73]:=
```
Reduce[DwPrOptimalContractprevious[m, beta, sR] - D[alphaP[m, beta, b, theta], b] ≤ 0 &&
  0 < beta < 1 && 0 < m < 1 / 2 && (1 - beta) ≥ 2 m beta &&
  m > sR > -m && overlineTheta[m, beta, b] < 1 / 2 &&
  0 < b < bHat[m, beta, theta] && theta > overlineTheta[m, beta, b]]
```

Out[73]=
```
False
```

In[74]:=
```
(* SANITY CHECK *)
(*Checking whether for no concession by R (because of no support)
  there is still concessions by L possible --- should return 0*)
```

In[75]:=
```
FullSimplify[With[{sR = -m}, wPrOptimalContractprevious[m, beta, b, theta, sR]] -
  alphaP[m, beta, b, theta]]
```

Out[75]=
```
0
```

In[76]:=
```
(*Solving for the sR such that P's enforcement constraint on-
 path binds also after R's lead*)
(* Trivially holds when sR=-m. We are looking for the second root*)
```

In[77]:=
```
solvesRmax = Simplify[
  Solve[wPrOptimalContractprevious[m, beta, b, theta, sR] == alphaP[m, beta, b, theta], sR]]
sRmax[m_, beta_, b_, theta_] = Simplify[sR /. solvesRmax[[2]]];
FullSimplify[D[sRmax[m, beta, b, theta], b]]
(*We show now that the second root is non-decreasing in b*)

Reduce[FullSimplify[D[sRmax[m, beta, b, theta], b]] < 0 && 0 < beta < 1 && 0 < m < 1 / 2 &&
   (1 - beta) ≥ 2 m beta && m > sR > -m && overlineTheta[m, beta, b] < 1 / 2 &&
  0 < b < bHat[m, beta, theta] && theta > overlineTheta[m, beta, b]]
```

Out[77]=

$$\{\{sR \to -m\}, \{sR \to (4 - 8\,theta + beta^2\,(2 + b + 8\,m + 6\,b\,m - 2\,theta - 4\,m\,theta) - 4\,beta\,(2 + b + m - 2\,theta - 2\,m\,theta)) / (2\,beta\,(-2 + b\,beta\,(-1 + 2\,m) + 4\,theta - 2\,beta\,(2\,m\,(-1 + theta) + theta)))\}\}$$

Out[79]=

$$-\frac{(-2 + beta + 2\,beta\,m)\,(-3 + beta + 2\,(1 + beta)\,m)\,(-1 + 2\,theta)}{(-2 + b\,beta\,(-1 + 2\,m) + 4\,theta - 2\,beta\,(2\,m\,(-1 + theta) + theta))^2}$$

Out[80]=
```
False
```

In[81]:=
```
(* Now we have to show that at this root, P is better off than at sR=-m*)
(* To do this, we need *)
```

In[82]:=
```
benefitsAfterLleads[m_, beta_, b_, theta_] =
 FullSimplify[wPlOptimalContractprevious[m, beta, b, theta, sRmax[m, beta, b, theta]] -
   alphaP[m, beta, b, theta]]
```

Out[82]=

$$((-4 + beta\,(1 + 2\,m)^2)\,(4 + (beta + 2\,beta\,m)^2\,(2 + b - 2\,theta) - 8\,theta - 4\,beta\,(2 + b + 2\,m - 2\,theta - 4\,m\,theta))) / (4\,beta\,(-2 + beta + 2\,beta\,m)\,(-3 + beta + 2\,(1 + beta)\,m))$$

In[83]:=
```
(* Determine derivative wrt to b*)
```

In[84]:=
```
DbenefitsAfterLleads[m_, beta_, b_, theta_] = D[benefitsAfterLleads[m, beta, b, theta], b]
```

Out[84]=

$$\frac{\left(-4+\text{beta}\ (1+2\ \text{m})^2\right)\ \left(-4\ \text{beta}+(\text{beta}+2\ \text{beta}\ \text{m})^2\right)}{4\ \text{beta}\ (-2+\text{beta}+2\ \text{beta}\ \text{m})\ (-3+\text{beta}+2\ (1+\text{beta})\ \text{m})}$$

In[85]:=
```
(*Show that this derivative is positive, i.e. better b make it more profitable*)
```

In[86]:=
```
Reduce[DbenefitsAfterLleads[m, beta, b, theta] < 0 && 0 < beta < 1 &&
  0 < m < 1 / 2 && (1 - beta) ≥ 2 m beta && overlineTheta[m, beta, b] < 1 / 2 &&
  0 < b < bHat[m, beta, theta] && theta > overlineTheta[m, beta, b]]
```

Out[86]=

```
False
```

In[87]:=
```
(* Now there are two potential cutoffs: Either benefitsAfterLleades=
 alphaP or sRmax=-1. We determine both. The max of them is bcheck *)
```

In[88]:=
```
bCheck1[m_, beta_, theta_] = b /. FullSimplify[Solve[sRmax[m, beta, b, theta] == -m, b]][[1]]
```

Out[88]=

$$\frac{-4+8\ \text{beta}\ (1+\text{m})-2\ (\text{beta}+2\ \text{beta}\ \text{m})^2+2\ (-2+\text{beta}+2\ \text{beta}\ \text{m})^2\ \text{theta}}{\text{beta}\ \left(-4+\text{beta}\ (1+2\ \text{m})^2\right)}$$

In[89]:=
```
bCheck2[m_, beta_, theta_] =
 b /. FullSimplify[Solve[benefitsAfterLleads[m, beta, b, theta] == 0, b]][[1]]
```

Out[89]=

$$\frac{-4+8\ \text{beta}\ (1+\text{m})-2\ (\text{beta}+2\ \text{beta}\ \text{m})^2+2\ (-2+\text{beta}+2\ \text{beta}\ \text{m})^2\ \text{theta}}{\text{beta}\ \left(-4+\text{beta}\ (1+2\ \text{m})^2\right)}$$

In[90]:=
```
(* Sanity check: They should coincide*)
FullSimplify[bCheck1[m, beta, theta] - bCheck2[m, beta, theta]]
```

Out[90]=

```
0
```

In[91]:=
```
(* SANITY CHECK *)

(*The bHat should coincide with wPrOptimalContractprevious==alphaP for sR=m*)
FullSimplify[
 (b /. FullSimplify[Solve[With[{sR = m}, wPrOptimalContractprevious[m, beta, b,
         theta, sR]] == alphaP[m, beta, b, theta], b]][[1]]) - bHat[m, beta, theta]]
```

Out[91]=

```
0
```