In[1]:=
```
ClearAll["Global`*"]

(*Probabilities of selection for the case sR is played and for the case sL*)

pR[sR_] = 1 / 2 + sR;
pL[sL_] = 1 / 2 + sL;


(*We compute the continuation values for strategies (sL, sR, xL, xR)*)

(* We use the following notation for the policies: we consider x_i,
the "concession" of an agent i,
hence L's policy y_L=  0 + x_L and R's policy y_R= 1 - x_R*)

U = Solve[
   wLl == (1 - pL[sL]) * (b - xL + beta * wLl ) + pL[sL] * (- (1 - xR) + beta * wLr) &&
    wLr == (1 - pR[sR]) * (b - xL + beta * wLl ) + pR[sR] * (- (1 - xR) + beta * wLr) &&
    wRl == (1 - pL[sL]) * (- (1 - xL) + beta * wRl ) + pL[sL] * (b - xR + beta * wRr) &&
    wRr == (1 - pR[sR]) * (- (1 - xL) + beta * wRl ) + pR[sR] * (b - xR + beta * wRr) &&
    wPl ==
     (1 - pL[sL]) * (- (theta - xL) + beta * wPl ) + pL[sL] * (- (1 - xR - theta) + beta * wPr) &&
    wPr == (1 - pR[sR]) * (- (theta - xL) + beta * wPl ) +
      pR[sR] * (- (1 - xR - theta) + beta * wPr), {wLl, wLr, wRl, wRr, wPl, wPr}];

(*R's continuation value when R was selected last*)
wRr[beta_, b_, sL_, sR_, xL_, xR_] = FullSimplify[ wRr /. U[[1]]];
(*L's continuation value when L was selected last*)
wLl[beta_, b_, sL_, sR_, xL_, xR_] = FullSimplify[ wLl /. U[[1]]];
(*Principal's continuation value when L was selected last*)
wPl[beta_, b_, theta_, sL_, sR_, xL_, xR_] = FullSimplify[ wPl /. U[[1]]];
(*Principal's continuation value when R was selected last*)
wPr[beta_, b_, theta_, sL_, sR_, xL_, xR_] = FullSimplify[ wPr /. U[[1]]];

(**Principal's and R's punishment**)

alphaR[m_, beta_, b_] = FullSimplify[ With[{xL = 0},
   With[{xR = 0}, With[{sL = -m}, With[{sR = -m}, wRr[beta, b, sL, sR, xL, xR]]]]]]
alphaP[m_, beta_, b_, theta_] = FullSimplify[ With[{xL = 0},
   With[{xR = 0}, With[{sL = -m}, With[{sR = -m}, wPl[beta, b, theta, sL, sR, xL, xR]]]]]]
(*At the end of the code we show this is the right punishment
 for the Principal in the region we are interested in here*)
```

Out[9]=

$$\frac{1 - b + 2\,(1 + b)\,m}{2\,(-1 + \text{beta})}$$

Out[10]=

$$\frac{1 + m\ (-2 + 4\ theta)}{2\ (-1 + beta)}$$

In[11]:=

```
(**Agent's enforcement constraints**)

enforcementLmax[m_, beta_, b_, sL_, sR_, xL_, xR_, alphaL_] =
  FullSimplify[ xL + beta * (alphaL - wLl) /. U[[1]]];
enforcementRmax[m_, beta_, b_, sL_, sR_, xL_, xR_] =
  FullSimplify[ xR + beta * (alphaR[m, beta, b] - wRr) /. U[[1]]];
```

In[13]:=

```
(**R' Actions in the Optimal contract**)
(*Optimal contract -- Solving R's enforcement constraints*)

enforcementRmaxsubstituted[m_, beta_, b_, xL_, xR_] = FullSimplify[
  With[{sR = m}, With[{sL = -m}, enforcementRmax[m, beta, b, sL, sR, xL, xR]]]]
solvexR = FullSimplify[Solve[enforcementRmaxsubstituted[m, beta, b, xL, xR] == 0, xR]]
xRmax[m_, beta_, b_, xL_] = FullSimplify[xR /. solvexR[[1]]];
```

Out[13]=

$$\frac{2\ (1 + b)\ beta\ m\ (-2 + beta + 2\ beta\ m) + beta\ (-1 + 2\ m)\ xL - (-2 + beta + 2\ beta\ m)\ xR}{2\ (-1 + beta)\ (-1 + 2\ beta\ m)}$$

Out[14]=

$$\left\{\left\{xR \to 2\ (1 + b)\ beta\ m + \frac{beta\ (-1 + 2\ m)\ xL}{-2 + beta + 2\ beta\ m}\right\}\right\}$$

In[16]:=

```
(**L' Punishment -- P's continuation value at the back-to-business phase**)

Carrot[m_, beta_, b_, theta_, xL_] =
 FullSimplify[With[{xR = xRmax[m, beta, b, xL]}, With[{sR = m},
   With[{sL = -m}, -(theta - xL) + beta * wPl[beta, b, theta, sL, sR, xL, xR]]]]]
```

Out[16]=

$$\Big((-2 + beta + 2\ beta\ m)\ (-beta\ (-1 + 2\ m)\ (-1 + 2\ (1 + b)\ beta\ m) + 2\ (-1 + beta)\ theta) - 2\left(2 + beta\left(-2 + beta - 4\ m + 4\ beta\ m^2\right)\right) xL\Big) \Big/ (2\ (-1 + beta)\ (-1 + 2\ beta\ m)\ (-2 + beta + 2\ beta\ m))$$

In[17]:=

```
(**L' Punishment -- P's continuation value at punishment phase**)

Stickprevious[m_, beta_, b_, theta_, xL_, xR_, Stick_] = FullSimplify[With[{sR = m},
   (1 - pR[sR]) * Carrot[m, beta, b, theta, xL] + pR[sR] * (- (1 - xR - theta) + beta * Stick)]]
sticksolve1 = Solve[Stick == Stickprevious[m, beta, b, theta, xL, xR, Stick], Stick]
Stick[m_, beta_, b_, theta_, xL_, xR_] = FullSimplify[Stick /. sticksolve1[[1]]];
(*Now we can solve for R's concession in the punishment phase of L's Punishment*)

sticksolve2 =
 FullSimplify[Solve[Stick[m, beta, b, theta, xL, xR] == alphaP[m, beta, b, theta], xR]]
xRstick[m_, beta_, b_, theta_, xL_] = FullSimplify[xR /. sticksolve2[[1]]];
```

Out[17]=

$$\left((-1+2\,m)\left((-2+beta+2\,beta\,m)\,(beta\,(-1+2\,m)\,(-1+2\,(1+b)\,beta\,m)-2\,(-1+beta)\,theta)+2\left(2+beta\left(-2+beta-4\,m+4\,beta\,m^2\right)\right)xL\right)\right)\Big/\left(4\,(-1+beta)\,(-1+2\,beta\,m)\,(-2+beta+2\,beta\,m)\right)+\left(\frac{1}{2}+m\right)(-1+beta\,Stick+theta+xR)$$

Out[18]=

$$\left\{\left\{Stick\to\frac{1}{1-beta\left(\frac{1}{2}+m\right)}\left(-\frac{1}{2}-m+\left(\frac{1}{2}+m\right)theta+\left((-1+2\,m)\left((-2+beta+2\,beta\,m)\,(beta\,(-1+2\,m)\,(-1+2\,(1+b)\,beta\,m)-2\,(-1+beta)\,theta)+2\left(2+beta\left(-2+beta-4\,m+4\,beta\,m^2\right)\right)xL\right)\right)\Big/\left(4\,(-1+beta)\,(-1+2\,beta\,m)\,(-2+beta+2\,beta\,m)\right)+\left(\frac{1}{2}+m\right)xR\right)\right\}\right\}$$

Out[20]=

$$\left\{\left\{xR\to\left(m\,(-2+beta+2\,beta\,m)\left(4-8\,theta+beta\left(-4-b\,beta\,(1-2\,m)^2+8\,(-1+beta)\,m-2\,(1+2\,m)\,(-4+beta+2\,beta\,m)\,theta\right)\right)-(-1+2\,m)\left(2+beta\left(-2+beta-4\,m+4\,beta\,m^2\right)\right)xL\right)\Big/\left((-1+beta)\,(1+2\,m)\,(-1+2\,beta\,m)\,(-2+beta+2\,beta\,m)\right)\right\}\right\}$$

In[22]:=
```
(*Once we have obtained R's concessions in the L's punihsment phase,
we can obtain L's punishment*)

(*First, we compute L's punishment value at the start of the back-to-business phase*)

LatCarrot[m_, beta_, b_, xL_] =
 FullSimplify[With[{xR = xRmax[m, beta, b, xL]}, With[{sR = m}, With[{sL = -m},
     b - xL + beta * wLl[beta, b, sL, sR, xL, xR]]]]]

(*Second, we compute L's punishment value at the start of the punishment phase *)

LatStickprevious[m_, beta_, b_, theta_, xL_, LatStick_] =
  FullSimplify[With[{sR = m}, With[{xR = xRstick[m, beta, b, theta, xL]},
     (1 - pR[sR]) * LatCarrot[m, beta, b, xL] + pR[sR] * (-(1 - xR) + beta * LatStick)]]];

sticksolve3 = FullSimplify[
  Solve[LatStick == LatStickprevious[m, beta, b, theta, xL, LatStick], LatStick]]

(*which gives us L's punishment as a function
 of L's concessions in the Optimal contract, xL*)

LatStick[m_, beta_, b_, theta_, xL_] = FullSimplify[LatStick /. sticksolve3[[1]]]
```

Out[22]=

$$\frac{b\ (-2 + \text{beta}) + \text{beta} - 2\ (1 + b)\ \text{beta}\ m}{2\ (-1 + \text{beta})} + \frac{2\ xL}{-2 + \text{beta} + 2\ \text{beta}\ m}$$

Out[24]=

$$\left\{\left\{\text{LatStick} \to \frac{-1 + b - 2\ b\ m - 2\ xL + 2\ m\ (1 + \text{beta} - 4\ \text{theta} + 2\ \text{beta}\ \text{theta} + \text{beta}\ m\ (-2 + 4\ \text{theta}) + 2\ xL)}{2\ (-1 + \text{beta})\ (-1 + 2\ \text{beta}\ m)}\right\}\right\}$$

Out[25]=

$$\frac{-1 + b - 2\ b\ m - 2\ xL + 2\ m\ (1 + \text{beta} - 4\ \text{theta} + 2\ \text{beta}\ \text{theta} + \text{beta}\ m\ (-2 + 4\ \text{theta}) + 2\ xL)}{2\ (-1 + \text{beta})\ (-1 + 2\ \text{beta}\ m)}$$

In[26]:=
```
(*Closing the Optimal contract and L's punishment
 together -- Finding L's concessions in the Optimal contract, xL*)

enforcementLmaxsubstituted[m_, beta_, b_, theta_, xL_] = FullSimplify[
  With[{alphaL = LatStick[m, beta, b, theta, xL]}, With[{xR = xRmax[m, beta, b, xL]},
    With[{sR = m}, With[{sL = -m}, enforcementLmax[m, beta, b, sL, sR, xL, xR, alphaL]]]]]]

xLmaxsolve =
 FullSimplify[Solve[enforcementLmaxsubstituted[m, beta, b, theta, xL] == 0, xL]]
xLmax[m_, beta_, b_, theta_] = FullSimplify[xL /. xLmaxsolve[[1]]];
```

Out[26]=

$$\frac{\text{beta m } (-2+\text{beta}+2\text{ beta m})^2\ (\text{b}+2\text{ theta})+(-2+\text{beta }(4+\text{beta }(-1+4\ (-1+\text{m})\ \text{m})))\ \text{xL}}{(-1+\text{beta})\ (-1+2\text{ beta m})\ (-2+\text{beta}+2\text{ beta m})}$$

Out[27]=

$$\left\{\left\{\text{xL}\to-\frac{\text{beta m }(-2+\text{beta}+2\text{ beta m})^2\ (\text{b}+2\text{ theta})}{-2+\text{beta }(4+\text{beta }(-1+4\ (-1+\text{m})\ \text{m}))}\right\}\right\}$$

In[29]:=
```
(**Obtaining L's Punishment and the Optimal Contract -- we substitute xL**)

alphaL[m_, beta_, b_, theta_] =
  FullSimplify[With[{xL = xLmax[m, beta, b, theta]}, LatStick[m, beta, b, theta, xL]]];
xRstickFinal[m_, beta_, b_, theta_] =
 FullSimplify[With[{xL = xLmax[m, beta, b, theta]}, xRstick[m, beta, b, theta, xL]]]
xRmaxFinal[m_, beta_, b_, theta_] =
 FullSimplify[With[{xL = xLmax[m, beta, b, theta]}, xRmax[m, beta, b, xL]]]
```

Out[30]=

$$(4\text{ m }(-2+4\text{ theta}+\text{beta }(4+\text{b }(-1+2\text{ m})\ (-1+2\text{ beta m})-6\text{ theta}-4\text{ m theta}+\text{beta }(-1+2\text{ theta}+4\text{ m }(-1+\text{m}+\text{theta})))))\ /\ ((1+2\text{ m})\ (-2+\text{beta }(4+\text{beta }(-1+4\ (-1+\text{m})\ \text{m}))))$$

Out[31]=

$$\text{beta m}\left(2\ (1+\text{b})-\frac{\text{beta }(-1+2\text{ m})\ (-2+\text{beta}+2\text{ beta m})\ (\text{b}+2\text{ theta})}{-2+\text{beta }(4+\text{beta }(-1+4\ (-1+\text{m})\ \text{m}))}\right)$$

In[32]:=
```
(**REGIONS -- Proof of PROPOSITIONS 4 AND 5**)

(* When are we at the commitment punishment? -- We obtain a thereshold in theta *)

FullSimplify[D[xRstickFinal[m, beta, b, theta], theta]]
(*monotone decreasing in theta*)
solvetheta1 = FullSimplify[Solve[xRstickFinal[m, beta, b, theta] == 0, theta]]
thetaOutofCommitmentOptimal[m_, beta_, b_] = FullSimplify[theta /. solvetheta1[[1]]];
(*We now show Proposition 4, the Optimal contract results hold for some theta < 1/2*)

Reduce[thetaOutofCommitmentOptimal[m, beta, b] ≥ 1 / 2 && 0 < beta < 1 && 0 < m < 1 / 2 && b > 0]
```

Out[32]=

$$\frac{8\,(-1+\text{beta})\,m\,(-2+\text{beta}+2\,\text{beta}\,m)}{(1+2\,m)\,(-2+\text{beta}\,(4+\text{beta}\,(-1+4\,(-1+m)\,m)))}$$

Out[33]=

$$\left\{\left\{\text{theta}\to\frac{2+\text{beta}\left(-4-b+\text{beta}+2\,(b+(2+b)\,\text{beta})\,m-4\,(1+b)\,\text{beta}\,m^2\right)}{2\,(-1+\text{beta})\,(-2+\text{beta}+2\,\text{beta}\,m)}\right\}\right\}$$

Out[35]=

False

In[36]:=
```
(*We can instead show our results with
 respect to b -- we obtain bBar from Proposition 3*)

solvebBar = FullSimplify[Solve[xRstickFinal[m, beta, b, theta] == 0, b]]
bBar[m_, beta_, theta_] = FullSimplify[b /. solvebBar[[1]]];
(*largest b such that punishment without
  concession is possible and commitment=no commitment*)
```

Out[36]=

$$\left\{\left\{b\to\frac{2-4\,\text{theta}+\text{beta}\,(-4+\text{beta}-2\,\text{beta}\,\text{theta}+(6+4\,m)\,\text{theta}-4\,\text{beta}\,m\,(-1+m+\text{theta}))}{\text{beta}\,(-1+2\,m)\,(-1+2\,\text{beta}\,m)}\right\}\right\}$$

In[38]:=
```
(*Sanity Check: The calculated shoudl be the same as inverting thetaHat*)
FullSimplify[
 (b /. Solve[thetaOutofCommitmentOptimal[m, beta, b] == theta, b][[1]]) - bBar[m, beta, theta]]
```

Out[38]=

0

In[39]:=
```
(*Now we compute \hat{b} from Proposition 5, that is,
the largest b such that P's on-path constraint is redundant*)
solvebHat =
 FullSimplify[Solve[xRstickFinal[m, beta, b, theta] == xRmaxFinal[m, beta, b, theta], b]]
bHat[m_, beta_, theta_] = FullSimplify[b /. solvebHat[[1]]];
(*largest b such that P's on-path constraint does not bind*)
```

Out[39]=

$$\left\{\left\{b \to \frac{4 + \text{beta}^2\ (2 + 8\ m\ (1 + m\ (-1 + \text{theta}))\ - 2\ \text{theta}) + 8\ \text{beta}\ (-1 + \text{theta}) - 8\ \text{theta}}{\text{beta}\ (4 + \text{beta}\ (-1 + 4\ (-2 + m)\ m))}\right\}\right\}$$

In[41]:=
```
(*We want now to compare this threshold with the theta
 (overline) at which the contract stops being (DEC,DEC)*)

(*To that end, we compute L's enforcement constraint in the Commitment-
 Optimal contract for the case in which L's concession equals theta*)

(*Recall that L's punishment value under commitment
  is idential to R's punishment value under non-commitment*)

enforcementLmaxsubstituted[m_, beta_, b_, theta_] =
  FullSimplify[With[{alphaL = alphaR[m, beta, b]},
    With[{xL = theta}, With[{xR = xRmax[m, beta, b, theta]}, With[{sR = m},
       With[{sL = -m}, enforcementLmax[m, beta, b, sL, sR, xL, xR, alphaL]]]]]]];

solvetheta2 =
 FullSimplify[Solve[enforcementLmaxsubstituted[m, beta, b, theta] == 0, theta]]
thetaoverline[m_, beta_, b_] = FullSimplify[ theta /. solvetheta2[[1]]];
```

Out[42]=

$$\left\{\left\{\text{theta} \to \frac{(1 + b)\ \text{beta}\ m\ (-2 + \text{beta} + 2\ \text{beta}\ m)}{-1 + \text{beta}}\right\}\right\}$$